

Avant!

Chapter 7

Performing Transient Analysis

Star-Hspice transient analysis computes the circuit solution as a function of time over a time range specified in the .TRAN statement.

This chapter covers the following topics:

- [Understanding the Simulation Flow](#)
- [Understanding Transient Analysis](#)
- [Using the .TRAN Statement](#)
- [Understanding the Control Options](#)
- [Testing for Speed, Accuracy and Convergence](#)
- [Selecting Timestep Control Algorithms](#)
- [Performing Fourier Analysis](#)
- [Using Data Driven PWL Sources](#)

Understanding the Simulation Flow

Figure 7-1 illustrates the transient analysis simulation flow for Star-Hspice.




Figure 7-1: Transient Analysis Simulation Flow

Understanding Transient Analysis

Since transient analysis is dependent on time, it uses different analysis algorithms, control options with different convergence-related issues and different initialization parameters than DC analysis. However, since a transient analysis first performs a DC operating point analysis (unless the UIC option is specified in the .TRAN statement), most of the DC analysis algorithms, control options, and initialization and convergence issues apply to transient analysis.

Initial Conditions for Transient Analysis

Some circuits, such as oscillators or circuits with feedback, do not have stable operating point solutions. For these circuits, either the feedback loop must be broken so that a DC operating point can be calculated or the initial conditions must be provided in the simulation input. The DC operating point analysis is bypassed if the UIC parameter is included in the .TRAN statement. If UIC is included in the .TRAN statement, a transient analysis is started using node voltages specified in an .IC statement. If a node is set to 5 V in a .IC statement, the value at that node for the first time point (time 0) is 5 V.

You can use the .OP statement to store an estimate of the DC operating point during a transient analysis.

Example

```
.TRAN 1ns 100ns UIC
.OP 20ns
```

The .TRAN statement UIC parameter in the above example bypasses the initial DC operating point analysis. The .OP statement calculates transient operating points at $t=0$ and $t=20$ ns during the transient analysis.

Although a transient analysis might provide a convergent DC solution, the transient analysis itself can still fail to converge. In a transient analysis, the error message “internal timestep too small” indicates that the circuit failed to converge. The convergence failure might be due to stated initial conditions that are not close enough to the actual DC operating point values.

Using the .TRAN Statement

Syntax

Single-point analysis:

```
.TRAN var1 START=start1 STOP=stop1 STEP=incr1
```

or

```
.TRAN var1 START=<param_expr1> STOP=<param_expr2>
+ STEP=<param_expr3>
```

Double-point analysis:

```
.TRAN var1 START=start1 STOP=stop1 STEP=incr1
+ <SWEEP var2 type np start2 stop2>
```

or

```
.TRAN tincr1 tstop1 <tincr2 tstop2 ...tincrN tstopN>
+ <START=val> <UIC> + <SWEEP var pstart
+ pstop pincr>
```

Parameterized sweep:

```
.TRAN tincr1 tstop1 <tincr2 tstop2 ...tincrN tstopN>
+ <START=val> <UIC>
```

Data driven sweep:

```
.TRAN DATA=datanm
```

or

```
.TRAN var1 START=start1 STOP=stop1 STEP=incr1
+ <SWEEP DATA=datanm>
```

or

```
.TRAN DATA=datanm<SWEEP var pstart pstop pincr>
```

Monte Carlo:

```
.TRAN tincr1 tstop1 <tincr2 tstop2 ...tincrN tstopN>
+ <START=val> <UIC><SWEEP MONTE=val>
```

Optimization:

```
.TRAN DATA=datanm OPTIMIZE=opt_par_fun
+ RESULTS=measnames MODEL=optmod
```

Transient sweep specifications can include the following keywords and parameters:

<i>DATA=datanm</i>	data name referred to in the .TRAN statement
<i>MONTE=val</i>	produces a number <i>val</i> of randomly generated values that are used to select parameters from a distribution. The distribution can be <i>Gaussian</i> , <i>Uniform</i> , or <i>Random Limit</i> .
<i>np</i>	number of points or number of points per decade or octave, depending on the preceding keyword
<i>param_expr...</i>	user-specified expressions—for example, param_expr1...param_exprN
<i>pincr</i>	voltage, current, element or model parameter, or temperature increment value Note: If “type” variation is used, the “np” (number of points) is specified instead of “pincr”.
<i>pstart</i>	starting voltage, current, temperature, any element or model parameter value Note: If type variation “POI” is used (list of points), a list of parameter values is specified instead of “pstart pstop”.
<i>pstop</i>	final voltage, current, temperature, any element or model parameter value
<i>START</i>	time at which printing or plotting is to begin. The START keyword is optional: you can specify start time without preceding it with “START=” Note: If the .TRAN statement is used in conjunction with a .MEASURE statement, using a nonzero START time can result in incorrect .MEASURE results. Nonzero START times should

	not be used in .TRAN statements when .MEASURE also is being used.
<i>SWEEP</i>	keyword to indicate a second sweep is specified on the .TRAN statement
<i>tincr1...</i>	printing or plotting increment for printer output, and the suggested computing increment for the postprocessor.
<i>tstop1...</i>	time at which the transient analysis stops incrementing by <i>tincr1</i> . If another <i>tincr-tstop</i> pair follows, the analysis continues with the new increment.
<i>type</i>	specifies any of the following keywords: DEC – decade variation OCT – octave variation (the value of the designated variable is eight times its previous value) LIN – linear variation POI – list of points
<i>UIC</i>	causes Star-Hspice to use the nodal voltages specified in the .IC statement (or by the “IC=” parameters in the various element statements) to calculate the initial transient conditions, rather than solving for the quiescent operating point
<i>var</i>	name of an independent voltage or current source, any element or model parameter, or the keyword TEMP (indicating a temperature sweep). Star-Hspice supports source value sweep, referring to the source name (SPICE style). However, if a parameter sweep, a .DATA statement, and a temperature sweep are specified, a parameter name must be chosen for the source value and subsequently referred to in the .TRAN statement. The parameter name must not start with V or I.

Examples

The following example performs and prints the transient analysis every 1 ns for 100 ns.

```
.TRAN 1NS 100NS
```

The following example performs the calculation every 0.1 ns for the first 25 ns, and then every 1 ns until 40 ns; the printing and plotting begin at 10 ns.

```
.TRAN .1NS 25NS 1NS 40NS START=10NS
```

The following example performs the calculation every 10 ns for 1 μ s; the initial DC operating point calculation is bypassed, and the nodal voltages specified in the .IC statement (or by IC parameters in element statements) are used to calculate initial conditions.

```
.TRAN 10NS 1US UIC
```

The following example increases the temperature by 10 °C through the range -55 °C to 75 °C and performs transient analysis for each temperature.

```
.TRAN 10NS 1US UIC SWEEP TEMP -55 75 10
```

The following performs an analysis for each load parameter value at 1 pF, 5 pF, and 10 pF.

```
.TRAN 10NS 1US SWEEP load POI 3 1pf 5pf 10pf
```

The following example is a data driven time sweep and allows a data file to be used as sweep input. If the parameters in the data statement are controlling sources, they must be referenced by a piecewise linear specification.

```
.TRAN data=dataname
```

Understanding the Control Options

The options in this section modify the behavior of the transient analysis integration routines. Delta refers to the internal timestep. TSTEP and TSTOP refer to the step and stop values entered with the .TRAN statement. The options are grouped into three categories: method, tolerance, and limit:

Method	Tolerance		Limit	
BYPASS	ABSH	RELH	AUTOSTOP	ITL3
CSHUNT	ABSV	RELI	BKPSIZ	ITL4
DVDT	ABSVAR	RELQ	DELMAX	ITL5
GSHUNT	ACCURATE	RELTOL	DVTR	RMAX
INTERP	BYTOL	RELV	FS	RMIN
ITRPRT	CHGTOL	RELVAR	FT	VFLOOR
LVLTIM	DI	SLOPETOL	GMIN	
MAXORD	FAST	TIMERES		
METHOD	MBYPASS	TRTOL		
	MAXAMP	VNTOL		
	MU	XMU		

Method Options

BYPASS speeds up simulation by not updating the status of latent devices. Setting .OPTION BYPASS=1 enables bypassing. BYPASS applies to MOSFETs, MESFETs, JFETs, BJTs, and diodes. Default=0.

Note: Use the BYPASS algorithm cautiously. For some types of circuits it can result in nonconvergence problems and loss of accuracy in transient analysis and operating point calculations.

CSHUNT capacitance added from each node to ground. Adding a small CSHUNT to each node can solve some “internal timestep too small” problems caused by high-frequency oscillations or numerical noise. Default=0.

DVDT allows the timestep to be adjusted based on node voltage rates of change. Choices are:

- 0 original algorithm
- 1 fast
- 2 accurate
- 3,4 balance speed and accuracy

Default=4.

The ACCURATE option also increases the accuracy of the results.

GSHUNT conductance added from each node to ground. The default value is zero. Adding a small GSHUNT to each node can solve some “internal timestep too small” problems caused by high frequency oscillations or by numerical noise.

INTERP limits output to post-analysis tools, such as Cadence or Zuken, to only the .TRAN timestep intervals. By default, Star-Hspice outputs all convergent iterations. INTERP typically produces a much smaller *design.tr#* file.

ITRPRT prints output variables at their internal timepoint values. Using this option can generate a long output list.

LVLTIM=x selects the timestep algorithm used for transient analysis. If LVLTIM=1, the DVDT timestep algorithm is used. If LVLTIM=2, the local truncation error timestep algorithm is used. If LVLTIM=3, the DVDT timestep algorithm with timestep reversal is used.

If the GEAR method of numerical integration and linearization is used, LVLTIM=2 is selected. If the TRAP linearization algorithm is used, LVLTIM 1 or 3 can be selected. Using LVLTIM=1 (the DVDT option) helps avoid the “internal timestep too small” nonconvergence problem. The local truncation algorithm (LVLTIM=2), however,

provides a higher degree of accuracy and prevents errors propagating from time point to time point, which can sometimes result in an unstable solution. Default=1.

MAXORD=x sets the maximum order of integration when the GEAR method is used (see METHOD). The value of x can be either 1 or 2. If MAXORD=1, the backward Euler method of integration is used. MAXORD=2, however, is more stable, accurate, and practical. Default=2.0.

METHOD=name sets the numerical integration method used for a transient analysis to either GEAR or TRAP. To use GEAR, set METHOD=GEAR. This automatically sets LVLTIM=2.

(You can change LVLTIM from 2 to 1 or 3 by setting LVLTIM=1 or 3 after the METHOD=GEAR option. This overrides the LVLTIM=2 setting made by METHOD=GEAR.)

TRAP (trapezoidal) integration generally results in reduced program execution time, with more accurate results. However, trapezoidal integration can introduce an apparent oscillation on printed or plotted nodes that might not be caused by circuit behavior. To test if this is the case, run a transient analysis with a small timestep. If the oscillation disappears, it was due to the trapezoidal method.

The GEAR method acts as a filter, removing the oscillations found in the trapezoidal method. Highly nonlinear circuits such as operational amplifiers can require very long execution times with the GEAR method. Circuits that are not convergent with trapezoidal integration often converge with GEAR. Default=TRAP (trapezoidal).

Tolerance Options

- ABSH*=*x* sets the absolute current change through voltage defined branches (voltage sources and inductors). In conjunction with DI and RELH, ABSH is used to check for current convergence. Default=0.0.
- ABSV*=*x* same as VNTOL. See VNTOL.
- ABSVAR*=*x* sets the limit on the maximum voltage change from one time point to the next. Used with the DVDT algorithm. If the simulator produces a convergent solution that is greater than ABSVAR, the solution is discarded, the timestep is set to a smaller value, and the solution is recalculated. This is called a timestep reversal. Default=0.5 (volts).
- ACCURATE** selects a time algorithm that uses LVLTIM=3 and DVDT=2 for circuits such as high-gain comparators. Circuits that combine high gain with large dynamic range should use this option to guarantee solution accuracy. When ACCURATE is set to 1, it sets the following control options:
 LVLTIM=3DVDT=2RELVAR=0.2
 ABSVAR=0.2FT=0.2RELMOS=0.01
 Default=0.
- BYTOL*=*x* specifies the tolerance for the voltage at which a MOSFET, MESFET, JFET, BJT, or diode is considered latent. Star-Hspice does not update the status of latent devices. Default=MBYPASS×VNTOL.
- CHGTOL*=*x* sets the charge error tolerance when LVLTIM=2 is set. CHGTOL, along with RELQ, sets the absolute and relative charge tolerance for all Star-Hspice capacitances. Default=1e-15 (coulomb).

- DI=x** sets the maximum iteration-to-iteration current change through voltage defined branches (voltage sources and inductors). This option is only applicable when the value of the DI control option is greater than 0. Default=0.0.
- FAST** speeds up simulation by not updating the status of latent devices. This option is applicable for MOSFETs, MESFETs, JFETs, BJTs, and diodes. Default=0.
- A device is considered to be latent when its node voltage variation from one iteration to the next is less than the value of either the BYTOL control option or the BYPASSTOL element parameter. (When FAST is on, Star-Hspice sets BYTOL to different values for different types of device models.)
- In addition to the FAST option, the input preprocessing time can be reduced by the options NOTOP and NOELCK. Increasing the value of the MBYPASS option or the BYTOL option setting also helps simulations run faster, but can reduce accuracy.
- MAXAMP=x** sets the maximum current through voltage defined branches (voltage sources and inductors). If the current exceeds the MAXAMP value, an error is issued. Default=0.0.
- MBYPASS=x** used to compute the default value for the BYTOL control option:
- $$\text{BYTOL} = \text{MBYPASS} \times \text{VNTOL}$$
- Also multiplies voltage tolerance RELV.
 MBYPASS should be set to about 0.1 for precision analog circuits.
 Default=1 for DVDT=0, 1, 2, or 3. Default=2 for DVDT=4.
- MU=x,** the coefficient for trapezoidal integration. The range for MU is 0.0 to 0.5.
- XMU=x** XMU is the same as MU. Default=0.5.

<i>RELH=x</i>	sets relative current tolerance through voltage defined branches (voltage sources and inductors). RELH is used to check current convergence. This option is applicable only if the value of the ABSH control option is greater than zero. Default=0.05.
<i>RELI=x</i>	sets the relative error/tolerance change, in percent, from iteration to iteration to determine convergence for all currents in diode, BJT, and JFET devices. (RELMOS sets the tolerance for MOSFETs). This is the percent change in current from the value calculated at the previous timestep. Default=1 (%) for KCLTEST=0, 1e-4 (%) for KCLTEST=1.
<i>RELQ=x</i>	used in the local truncation error timestep algorithm (LVLTIM=2). RELQ changes the size of the timestep. If the capacitor charge calculation of the present iteration exceeds that of the past iteration by a percentage greater than the value of RELQ, the internal timestep (Delta) is reduced. Default=0.01 (1%).
<i>RELTOL,</i> <i>RELV</i>	sets the relative error tolerance for voltages. RELV is used in conjunction with the ABSV control option to determine voltage convergence. Increasing RELV increases the relative error. RELV is the same as RELTOL. Options RELI and RELVDC default to the RELTOL value. Default=1e-3.
<i>RELVAR=x</i>	used with ABSVAR and the timestep algorithm option DVDT. RELVAR sets the relative voltage change for LVLTIM=1 or 3. If the nodal voltage at the current time point exceeds the nodal voltage at the previous time point by RELVAR, the timestep is reduced and a new solution at a new time point is calculated. Default=0.30 (30%).
<i>SLOPETOL=x</i>	sets a lower limit for breakpoint table entries in a piecewise linear (PWL) analysis. If the difference in the slopes of two consecutive PWL segment is less than the SLOPETOL value, the breakpoint table entry for the point between the segments is ignored. Default=0.5.

<i>TIMERES</i> = <i>x</i>	sets a minimum separation between breakpoint values for the breakpoint table. If two breakpoints are closer together in time than the <i>TIMERES</i> value, only one of them is entered in the breakpoint table. Default=1 ps.
<i>TRTOL</i> = <i>x</i>	used in the local truncation error timestep algorithm (<i>LVLTIM</i> =2). <i>TRTOL</i> is a multiplier of the internal timestep generated by the local truncation error timestep algorithm. <i>TRTOL</i> reduces simulation time, while maintaining accuracy. It is a factor that estimates the amount of error introduced by truncating the Taylor series expansion used in the algorithm. This error is a reflection of what the minimum value of the timestep should be to reduce simulation time and maintain accuracy. The range of <i>TRTOL</i> is 0.01 to 100, with typical values being in the 1 to 10 range. If <i>TRTOL</i> is set to 1, the minimum value, a very small timestep is used. As the setting of <i>TRTOL</i> increases, the timestep size increases. Default=7.0.
<i>VNTOL</i> = <i>x</i> , <i>ABSV</i>	sets the absolute minimum voltage for DC and transient analysis. Decrease <i>VNTOL</i> if accuracy is of more concern than convergence. If voltages less than 50 microvolts are required, <i>VNTOL</i> can be reduced to two orders of magnitude less than the smallest desired voltage, ensuring at least two digits of significance. Typically, <i>VNTOL</i> need not be changed unless the circuit is a high voltage circuit. For 1000 volt circuits, a reasonable value can be 5 to 50 millivolts. <i>ABSV</i> is the same as <i>VNTOL</i> . Default=50 (microvolts).
<i>XMU</i> = <i>x</i>	Same as <i>MU</i> . See <i>MU</i> .

Limit Options

<i>AUTOSTOP</i>	stops the transient analysis when all <i>TRIG-TARG</i> and <i>FIND-WHEN</i> measure functions are calculated. This option can result in a substantial CPU time reduction. If the data file
-----------------	--

contains measure functions such as AVG, RMS, MIN, MAX, PP, ERR, ERR1,2,3, and PARAM, then AUTOSTOP is disabled.

- BKPSIZ=x* sets the size of the breakpoint table. Default=5000.
- DELMAX=x* sets the maximum value for the internal timestep Delta. Star-Hspice automatically sets the DELMAX value based on various factors, which are listed in *Timestep Control for Accuracy, page -19 in , Performing Transient Analysis*. This means that the initial DELMAX value shown in the Star-Hspice output listing is generally not the value used for simulation.
- DVTR* allows the use of voltage limiting in transient analysis. Default=1000.
- FS=x* sets the fraction of a timestep (TSTEP) that Delta (the internal timestep) is decreased for the first time point of a transient. Decreasing the FS value helps circuits that have timestep convergence difficulties. It also is used in the DVDT=3 method to control the timestep.
- $$Delta = FS \times [MIN(TSTEP, DELMAX, BKPT)]$$
- where DELMAX is specified and BKPT is related to the breakpoint of the source. TSTEP is set in the .TRAN statement. Default=0.25.
- FT=x* sets the fraction of a timestep (TSTEP) by which Delta (the internal timestep) is decreased for an iteration set that does not converge. It is also used in DVDT=2 and DVDT=4 to control the timestep. Default=0.25.
- GMIN=x* sets the minimum conductance allowed for in a transient analysis time sweep. Default=1e-12.
- IMAX=x,* determines the maximum timestep in the timestep

- ITL4*=*x* algorithms used for transient analysis simulations. IMAX sets an upper limit on the number of iterations allowed to obtain a convergent solution at a timepoint. If the number of iterations needed is greater than IMAX, the internal timestep Delta is decreased by a factor equal to the transient control option FT, and a new solution is calculated using the new timestep. IMAX also works in conjunction with the transient control option IMIN. ITL4 is the same as IMAX. Default=8.0.
- IMIN*=*x*,
ITL3=*x* determines the timestep in the algorithms used for transient analysis simulations. IMIN sets a lower limit on the number of iterations required to obtain convergence. If the number of iterations is less than IMIN, the internal timestep, Delta, is doubled. This option is useful for decreasing simulation times in circuits where the nodes are stable most of the time, such as digital circuits. If the number of iterations is greater than IMIN, the timestep is kept the same unless the option IMAX is exceeded (see IMAX). ITL3 is the same as IMIN. Default=3.0.
- ITL3*=*x* same as IMIN. See IMIN.
- ITL4*=*x* same as IMAX. See IMAX.
- ITL5*=*x* sets the transient analysis total iteration limit. If a circuit uses more than ITL5 iterations, the program prints all results to that point. The default allows an infinite number of iterations. Default=0.0.
- RMAX*=*x* sets the TSTEP multiplier, which determines the maximum value, DELMAX, that can be used for the internal timestep Delta:
$$\text{DELMAX} = \text{TSTEP} \times \text{RMAX}$$
Default=5 when dvdt=4 and lvltim=1, otherwise, default=2.
- RMIN*=*x* sets the minimum value of Delta (internal timestep). An internal timestep smaller than $\text{RMIN} \times \text{TSTEP}$ results in termination of the transient analysis with the error message

“internal timestep too small”. Delta is decreased by the amount set by the FT option if the circuit has not converged in IMAX iterations. Default=1.0e-9.

VFLOOR=*x* sets a lower limit for the voltages that are printed in the output listing. All voltages lower than VFLOOR are printed as 0. This only affects the output listing; the minimum voltage used in a simulation is set by VNTOL (ABSV).

Matrix Manipulation Options

After linearization of the individual elements within a Star-Hspice input netlist file, the linear equations are constructed for the matrix. User-controlled variables affecting the construction and solution of the matrix equation include options PIVOT and GMIN. GMIN places a variable into the matrix that prevents the matrix becoming ill-conditioned.

Pivot Option

Select the PIVOT option for a number of different pivoting methods to reduce simulation time and assist in both DC and transient convergence. Pivoting reduces the error resulting from elements in the matrix that are widely different in magnitude. The use of PIVOT results in a search of the matrix for the largest element value. This element value then is used as the pivot.

Testing for Speed, Accuracy and Convergence

Convergence is defined as the ability to obtain a solution to a set of circuit equations within a given tolerance criteria and number of iterations. In numerical circuit simulation, the designer specifies a relative and absolute accuracy for the circuit solution and the simulator iteration algorithm attempts to converge to a solution that is within these set tolerances. In many cases the speed of reaching a solution also is of primary interest to the designer, particularly for preliminary design trials, and some accuracy is willingly sacrificed.

Simulation Speed

Star-Hspice can substantially reduce the computer time needed to solve complex problems. The following user options alter internal algorithms to increase simulation efficiency.

- .OPTIONS FAST – sets additional options that increase simulation speed with little loss of accuracy
- .OPTIONS AUTOSTOP – terminates the simulation when all .MEASURE statements have completed. This is of special interest when testing corners.

The FAST and AUTOSTOP options are described in “Understanding the Control Options” which starts on page 7-8.

Simulation Accuracy

Star-Hspice is shipped with control option default values that aim for superior accuracy while delivering good performance in simulation time. The control options and their default settings to maximize accuracy are:

DVDT=4 LVLTIM=1 RMAX=5 SLOPETOL=0.75
FT=FS=0.25 BYPASS=1 BYTOL=MBYPASSxVNTOL=0.100m

Note: BYPASS is only turned on (set to 1) when DVDT=4. For other DVDT settings, BYPASS is off (0). SLOPETOL is set to 0.75 when DVDT=4 and LVLTIM=1. For all other values of DVDT or LVLTIM, SLOPETOL defaults to 0.5.

Timestep Control for Accuracy

The DVDT control option selects the timestep control algorithm. Relationships between DVDT and other control options are discussed in [Selecting Timestep Control Algorithms, page -24](#) in , *Performing Transient Analysis*.

The DELMAX control option also affects simulation accuracy. DELMAX specifies the maximum allowed timestep size. If DELMAX is not set in an .OPTIONS statement, Star-Hspice computes a DELMAX value. Factors that determine the computed DELMAX value are:

- .OPTIONS RMAX and FS
- Breakpoint locations for a PWL source
- Breakpoint locations for a PULSE source
- Smallest period for a SIN source
- Smallest delay for a transmission line component
- Smallest ideal delay for a transmission line component
- TSTEP value in a .TRAN analysis
- Number of points in an FFT analysis

The FS and RMAX control options provide some user control over the DELMAX value. The FS option, which defaults to 0.25, scales the breakpoint interval in the DELMAX calculation. The RMAX option, which defaults to 5 if DVDT=4 and LVLTIM=1, scales the TSTEP (timestep) size in the DELMAX calculation.

For circuits that contain oscillators or ideal delay elements, an .OPTIONS statement should be used to set DELMAX to one-hundredth of the period or less.

The ACCURATE control option tightens the simulation options to give the most accurate set of simulation algorithms and tolerances. When ACCURATE is set to 1, it sets the following control options:

DVDT=2	LVLTIM=3	FT=FS=0.2	SLOPETOL=0.5
BYTOL=0	BYPASS=0	RMAX=2	
RELVAR=0.2	ABSVAR=0.2	RELMOS=0.01	

Models and Accuracy

Simulation accuracy relies heavily on the sophistication and accuracy of the models used. More advanced MOS, BJT, and GaAs models give superior results for critical applications. Simulation accuracy is increased by:

- Algebraic models that describe parasitic interconnect capacitances as a function of the width of the transistor. The wire model extension of the resistor can model the metal, diffusion, or poly interconnects to preserve the relationship between the physical layout and electrical property.
- MOS model parameter ACM that calculates defaults for source and drain junction parasitics. Star-Hspice uses ACM equations to calculate the size of the bottom wall, the length of the sidewall diodes, and the length of a lightly doped structure. SPICE defaults with no calculation of the junction diode. Specify AD, AS, PD, PS, NRD, NRS to override the default calculations.
- MOS model parameter CAPOP=4 that models the most advanced charge conservation, non-reciprocal gate capacitances. The gate capacitors and overlaps are calculated from the IDS model.

Guidelines for Choosing Accuracy Options

Use the ACCURATE option for

- Analog or mixed signal circuits
- Circuits with long time constants, such as RC networks
- Circuits with ground bounce

Use the default options (DVDT=4) for

- Digital CMOS
- CMOS cell characterization
- Circuits with fast moving edges (short rise and fall times)

For ideal delay elements, use one of the following:

- ACCURATE
- DVDT=3
- DVDT=4, and, if the minimum pulse width of any signal is less than the minimum ideal delay, set DELMAX to a value smaller than the minimum pulse width

Numerical Integration Algorithm Controls

When using Star-Hspice for transient analysis, you can select one of two options, Gear or Trapezoidal, to convert differential terms into algebraic terms.

Syntax

Gear algorithm:

```
.OPTION METHOD=GEAR
```

Trapezoidal algorithm (default):

```
.OPTION METHOD=TRAP
```

Each of these algorithms has advantages and disadvantages, but the trapezoidal is the preferred algorithm overall because of its highest accuracy level and lowest simulation time.

The selection of the algorithm is not, however, an elementary task. The appropriate algorithm for convergence depends to a large degree on the type of circuit and its associated behavior for different input stimuli.

Gear and Trapezoidal Algorithms

The timestep control algorithm is automatically set by the choice of algorithm. In Star-Hspice, if the GEAR algorithm is selected, the timestep control algorithm defaults to the truncation timestep algorithm. On the other hand, if the trapezoidal algorithm is selected, the DVDT algorithm is the default. You can change these Star-Hspice default by using the timestep control options.




Figure 7-2: Time Domain Algorithm




Figure 7-3: Iteration Algorithm

One limitation of the trapezoidal algorithm is that it can result in computational oscillation—that is, an oscillation caused by the trapezoidal algorithm and not by the circuit design. This also produces an unusually long simulation time. When this occurs in circuits that are inductive in nature, such as switching regulators, use the GEAR algorithm.

Selecting Timestep Control Algorithms

Star-Hspice allows the selection of three dynamic timestep control algorithms:

- Iteration count
- Truncation
- DVDT

Each of these algorithms uses a dynamically changing timestep. A dynamically changing timestep increases the accuracy of simulation and reduces the simulation time by varying the value of the timestep over the transient analysis sweep depending upon the stability of the output. Dynamic timestep algorithms increase the timestep value when internal nodal voltages are stable and decrease the timestep value when nodal voltages are changing quickly.




Figure 7-4: Internal Variable Timestep

In Star-Hspice, the timestep algorithm is selected by the LVLTIM option:

- LVLTIM=0 selects the iteration count algorithm.
- LVLTIM=1 selects the DVDT timestep algorithm, along with the iteration count algorithm. Operation of the timestep control algorithm is controlled by the setting of the DVDT control option. For LVLTIM=1 and DVDT=0, 1, 2, or 3, the algorithm does not use timestep reversal. For DVDT=4, the algorithm uses timestep reversal.

The DVDT algorithm is discussed further in *DVDT Dynamic Timestep Algorithm, page -26 in , Performing Transient Analysis.*

- LVLTIM=2 selects the truncation timestep algorithm, along with the iteration count algorithm with reversal.
- LVLTIM=3 selects the DVDT timestep algorithm with timestep reversal, along with the iteration count algorithm. For LVLTIM=3 and DVDT=0, 1, 2, 3, 4, the algorithm uses timestep reversal.

Iteration Count Dynamic Timestep Algorithm

The simplest dynamic timestep algorithm used is the iteration count algorithm. The iteration count algorithm is controlled by the following options:

IMAX	Controls the internal timestep size based on the number of iterations required for a timepoint solution. If the number of iterations per timepoint exceeds the IMAX value, the internal timestep is decreased. Default=8.
IMIN	Controls the internal timestep size based on the number of iterations required for the previous timepoint solution. If the last timepoint solution took fewer than IMIN iterations, the internal timestep is increased. Default=3.

Local Truncation Error (LTE) Dynamic Timestep Algorithm

The local truncation error timestep method uses a Taylor series approximation to calculate the next timestep for a transient analysis. This method calculates an internal timestep using the allowed local truncation error. If the calculated timestep is smaller than the current timestep, then the timepoint is set back (timestep reversal) and the calculated timestep is used to increment the time. If the calculated timestep is larger than the current one, then there is no need for a reversal. A new timestep is used for the next timepoint.

The local truncation error timestep algorithm is selected by setting LVLTIM=2.

The control options available with the local truncation error algorithm are:

TRTOL (default=7)
CHGTOL (default=1e-15)
RELQ (default=0.01)

DVDT Dynamic Timestep Algorithm

Select this algorithm by setting the option LVLTIM to 1 or 3. If you set LVLTIM=1, the DVDT algorithm does not use timestep reversal. The results for the current timepoint are saved, and a new timestep is used for the next timepoint. If you set LVLTIM=3, the algorithm uses timestep reversal. If the results are not converging at a given iteration, the results of current timepoint are ignored, time is set back by the old timestep, and a new timestep is used. Therefore, LVLTIM=3 is more accurate and more time consuming than LVLTIM=1.

The test the algorithm uses for reversing the timestep depends on the DVDT control option setting. For DVDT=0, 1, 2, or 3, the decision is based on the SLOPETOL control option value. For DVDT=4, the decision is based on the settings of the SLOPETOL, RELVAR, and ABSVAR control options.

The DVDT algorithm calculates the internal timestep based on the rate of nodal voltage changes. For circuits with rapidly changing nodal voltages, the DVDT algorithm uses a small timestep. For circuits with slowly changing nodal voltages, the DVDT algorithm uses larger timesteps.

The DVDT=4 setting selects a timestep control algorithm that is based on nonlinearity of node voltages, and employs timestep reversals if the LVLTIM option is set to either 1 or 3. The nonlinearity of node voltages is measured through changes in slopes of the voltages. If the change in slope is larger than the setting of the SLOPETOL control option, the timestep is reduced by a factor equal to the setting of the FT control option. The FT option defaults to 0.25. Star-Hspice sets the SLOPETOL value to 0.75 for LVLTIM=1, and to 0.50 for LVLTIM=3. Reducing the value of SLOPETOL increases simulation accuracy, but also increases simulation time. For LVLTIM=1, the simulation accuracy can be controlled by SLOPETOL and FT. For LVLTIM=3, the RELVAR and ABSVAR control options also affect the timestep, and therefore affect the simulation accuracy.

You can use options RELVAR and ABSVAR in conjunction with the DVDT option to improve simulation time or accuracy. For faster simulation time, RELVAR and ABSVAR should be increased (although this might decrease accuracy).

Note: If you need backward compatibility with Star-Hspice Release 95.3, use the following option values. Setting .OPTIONS DVDT=3 sets all of these values automatically.

```
LVLTIM=1   RMAX=2   SLOPETOL=0.5
FT=FS=0.25  BYPASS=0  BYTOL=0.050m
```

User Timestep Controls

The RMIN, RMAX, FS, FT, and DELMAX control options allow you to control the minimum and maximum internal timestep allowed for the DVDT algorithm. If the timestep falls below the minimum timestep default, the execution of the program halts. For example, an “internal timestep too small” error results when the timestep becomes less than the minimum internal timestep found by TSTEP×RMIN.

Note: RMIN is the minimum timestep coefficient and has a default value of 1e-9. TSTEP is the time increment, and is set in the .TRAN statement.

If DELMAX is set in an .OPTIONS statement, then DVDT=0 is used.

If DELMAX is not specified in an .OPTIONS statement, Star-Hspice computes a DELMAX value. For DVDT=0, 1, or 2, the maximum internal timestep is

$$\min[(TSTOP/50), DELMAX, (TSTEP \times RMAX)]$$


The TSTOP time is the transient sweep range set in the .TRAN statement.

In circuits with piecewise linear (PWL) transient sources, the SLOPETOL option also affects the internal timestep. A PWL source with a large number of voltage or current segments contributes a correspondingly large number of entries to the internal breakpoint table. The number of breakpoint table entries that must be considered contributes to the internal timestep control.


If the difference in the slope of consecutive segments of a PWL source is less than the SLOPETOL value, the breakpoint table entry for the point between the segments is ignored. For a PWL source with a signal that changes value slowly, ignoring its breakpoint table entries can help reduce the simulation time. Since the data in the breakpoint table is a factor in the internal timestep control, reducing the number of usable breakpoint table entries by setting a high SLOPETOL reduces the simulation time.

Performing Fourier Analysis

This section illustrates the flow for Fourier and FFT analysis.



.FOUR Statement



.FFT Statement

Figure 7-5: Fourier and FFT Analysis

.FOUR Statement

This statement performs a Fourier analysis as a part of the transient analysis. The Fourier analysis is performed over the interval (tstop-fperiod, tstop), where tstop is the final time specified for the transient analysis (see .TRAN statement), and fperiod is one period of the fundamental frequency (parameter “freq”). Fourier analysis is performed on 101 points of transient analysis data on the last 1/f time period, where f is the fundamental Fourier frequency. Transient data is interpolated to fit on 101 points running from (tstop-1/f) to tstop. The phase, the normalized component, and the Fourier component are calculated using 10 frequency bins. The Fourier analysis determines the DC component and the first nine AC components.

Syntax

```
.FOUR freq ov1 <ov2 ov3 ...>
```

freq the fundamental frequency

ov1 ... the output variables for which the analysis is desired

Example

```
.FOUR 100K V(5)
```

Accuracy and DELMAX

For maximum accuracy, .OPTION DELMAX should be set to (period/100). For circuits with very high resonance factors (high Q circuits such as crystal oscillators, tank circuits, and active filters) DELMAX should be set to less than (period/100).

Fourier Equation

The total harmonic distortion is the square root of the sum of the squares of the second through the ninth normalized harmonic, times 100, expressed as a percent:

$$THD = \frac{1}{R1} \cdot \left(\sum_{m=2}^9 R_m^2 \right)^{1/2} \cdot 100\%$$

This interpolation can result in various inaccuracies. For example, if the transient analysis runs at intervals longer than $1/(101 \cdot f)$, the frequency response of the interpolation dominates the power spectrum. Furthermore, there is no error range derived for the output.

The Fourier coefficients are calculated from:

$$g(t) = \sum_{m=0}^9 C_m \cdot \cos(mt) + \sum_{m=0}^9 D_m \cdot \sin(mt)$$

where

$$C_m = \frac{1}{\pi} \cdot \int_{-\pi}^{\pi} g(t) \cdot \cos(m \cdot t) \cdot dt$$

$$D_m = \frac{1}{\pi} \cdot \int_{-\pi}^{\pi} g(t) \cdot \sin(m \cdot t) \cdot dt$$

$$g(t) = \sum_{m=0}^9 C_m \cdot \cos(m \cdot t) + \sum_{m=0}^9 D_m \cdot \sin(m \cdot t)$$

C and D are approximated by:

$$C_m = \sum_{n=0}^{101} g(n \cdot \Delta t) \cdot \cos\left(\frac{2 \cdot \pi \cdot m \cdot n}{101}\right)$$

$$D_m = \sum_{n=0}^{101} g(n \cdot \Delta t) \cdot \sin\left(\frac{2 \cdot \pi \cdot m \cdot n}{101}\right)$$

The magnitude and phase are calculated by:

$$R_m = (C_m^2 + D_m^2)^{1/2}$$

$$\Phi_m = \arctan\left(\frac{C_m}{D_m}\right)$$

Example

The following is Star-Hspice input for an .OP, .TRAN, and .FOUR analysis.

```
CMOS INVERTER
*
M1 2 1 0 0 NMOS W=20U L=5U
M2 2 1 3 3 PMOS W=40U L=5U
VDD 3 0 5
VIN 1 0 SIN 2.5 2.5 20MEG
.MODEL NMOS NMOS LEVEL=3 CGDO=.2N CGSO=.2N CGBO=2N
.MODEL PMOS PMOS LEVEL=3 CGDO=.2N CGSO=.2N CGBO=2N
.OP
.TRAN 1N 100N
.FOUR 20MEG V(2)
.PRINT TRAN V(2) V(1)
.END
```

Output for the Fourier analysis is shown below.

```
*****
cmos inverter
***** fourier analysis                tnom= 25.000 temp=
25.000
*****
```



```

fourier components of transient response v(2)
dc component = 2.430D+00
harmonic frequency fourier normalized phase
normalized
no (hz) component component (deg) phase
(deg)
1 20.0000x 3.0462 1.0000 176.5386 0.
2 40.0000x 115.7006m 37.9817m -106.2672 -282.8057
3 60.0000x 753.0446m 247.2061m 170.7288 -5.8098
4 80.0000x 77.8910m 25.5697m -125.9511 -302.4897
5 100.0000x 296.5549m 97.3517m 164.5430 -11.9956
6 120.0000x 50.0994m 16.4464m -148.1115 -324.6501
7 140.0000x 125.2127m 41.1043m 157.7399 -18.7987
8 160.0000x 25.6916m 8.4339m 172.9579 -3.5807
9 180.0000x 47.7347m 15.6701m 154.1858 -22.3528

total harmonic distortion = 27.3791 percent

```

For further information on Fourier analysis, see [Chapter 25, Performing FFT Spectrum Analysis](#).

.FFT Statement

The syntax of the .FFT statement is shown below. The parameters are described in Table 7-1:

```

.FFT <output_var> <START=value> <STOP=value> <NP=value>
+ <FORMAT=keyword> <WINDOW=keyword> <ALFA=value> <FREQ=value>
+ <FMIN=value> <FMAX=value>

```

Table 7-1: .FFT Statement Parameters

Parameter	Default	Description
output_var	—	can be any valid output variable, such as voltage, current, or power
START	see Description	specifies the beginning of the output variable waveform to be analyzed. Defaults to the START value in the .TRAN statement, which defaults to 0.
FROM	see START	In .FFT statements, FROM is an alias for START.
STOP	see Description	specifies the end of the output variable waveform to be analyzed. Defaults to the TSTOP value in the .TRAN statement.
TO	see STOP	In .FFT statements, TO is an alias for STOP.
NP	1024	specifies the number of points used in the FFT analysis. NP must be a power of 2. If NP is not a power of 2, Star-Hspice automatically adjusts it to the closest higher number that is a power of 2.
FORMAT	NORM	specifies the output format: NORM=normalized magnitude UNORM=unnormalized magnitude
WINDOW	RECT	specifies the window type to be used: RECT=simple rectangular truncation window BART=Bartlett (triangular) window HANN=Hanning window HAMM=Hamming window BLACK=Blackman window HARRIS=Blackman-Harris window GAUSS=Gaussian window KAISER=Kaiser-Bessel window
ALFA	3.0	specifies the parameter used in GAUSS and KAISER windows to control the highest side-lobe level, bandwidth, and so on. 1.0 <= ALFA <= 20.0

Table 7-1: .FFT Statement Parameters

Parameter	Default	Description
FREQ	0.0 (Hz)	specifies a frequency of interest. If FREQ is nonzero, the output listing is limited to the harmonics of this frequency, based on FMIN and FMAX. The THD for these harmonics also is printed.
FMIN	1.0/T (Hz)	specifies the minimum frequency for which FFT output is printed in the listing file, or which is used in THD calculations. T = (STOP-START)
FMAX	0.5*NP*FMIN (Hz)	specifies the maximum frequency for which FFT output is printed in the listing file, or which is used in THD calculations.

.FFT Statement Syntax Examples

Below are four examples of valid .FFT statements.

```
.fft v(1)
.fft v(1,2) np=1024 start=0.3m stop=0.5m freq=5.0k
+ window=kaiser alfa=2.5
.fft I(rload) start=0m to=2.0m fmin=100k fmax=120k
+ format=unorm
.fft 'v(1) + v(2)' from=0.2u stop=1.2u window=harris
```

Only one output variable is allowed in a .FFT command. The following is an *incorrect* use of the command.

```
.fft v(1) v(2) np=1024
```

The correct use of the command is shown in the example below. In this case, a *.ft0* and a *.ft1* file are generated for the FFT of v(1) and v(2), respectively.

```
.fft v(1) np=1024
.fft v(2) np=1024
```

FFT Analysis Output

The results of the FFT analysis are printed in a tabular format in the *.lis* file, based on the parameters in the *.FFT* statement. The normalized magnitude values are printed unless you specify *FORMAT=UNORM*, in which case unnormalized magnitude values are printed. The number of printed frequencies is half the number of points (*NP*) specified in the *.FFT* statement. If you specify a minimum or a maximum frequency, using *FMIN* or *FMAX*, the printed information is limited to the specified frequency range. Moreover, if you specify a frequency of interest using *FREQ*, then the output is limited to the harmonics of this frequency, along with the percent of total harmonic distortion.

A *.ft#* file is generated, in addition to the listing file, for each FFT output variable, which contains the graphic data needed to display the FFT analysis waveforms. The magnitude in dB and the phase in degrees are available for display.

In the sample FFT analysis *.lis* file output below, notice that all the parameters used in the FFT analysis are defined in the header.

```
***** Sample FFT output extracted from the .lis file

fft test ... sine
*****  fft analysis                      tnom=  25.000
temp=  25.000
*****
fft components of transient response v(1)

Window: Rectangular
First Harmonic:    1.0000k
Start Freq:       1.0000k
Stop  Freq:       10.0000k

dc component: mag(db)= -1.132D+02  mag=  2.191D-06
phase=  1.800D+02

frequency  frequency  fft_mag  fft_mag  fft_phase
index      (hz)      (db)
  2         1.0000k      0.         1.0000    -3.8093m
  4         2.0000k -125.5914   525.3264n -5.2406
  6         3.0000k -106.3740   4.8007u   -98.5448
  8         4.0000k -113.5753   2.0952u   -5.5966
 10        5.0000k -112.6689   2.3257u   -103.4041
```

```

12          6.0000k   -118.3365      1.2111u   167.2651
14          7.0000k   -109.8888      3.2030u  -100.7151
16          8.0000k   -117.4413      1.3426u   161.1255
18          9.0000k    -97.5293     13.2903u   70.0515
20         10.0000k   -114.3693      1.9122u  -12.5492
total harmonic distortion =      1.5065m percent

```

The preceding example specifies a frequency of 1 kHz and the THD up to 10 kHz, which corresponds to the first ten harmonics.

Note: The highest frequency shown in the Star-Hspice FFT output might not be identical to the specified FMAX, due to Star-Hspice adjustments.

Table 7-2: describes the output of the Star-Hspice FFT analysis.

Table 7-2: .FFT Output Description

Column Heading	Description
frequency index	runs from 1 to NP/2, or the corresponding index for FMIN and FMAX. The DC component corresponding to index 0 is displayed independently.
frequency	the actual frequency associated with the index
fft_mag (db), fft_mag	there are two FFT magnitude columns: the first in dB and the second in the units of the output variable. The magnitude is normalized unless UNORM format is specified.
fft_phase	the associated phase, in degrees

Notes:

1. The following formula should be used as a guideline when specifying a frequency range for FFT output:

$$\text{frequency increment} = 1.0/(\text{STOP} - \text{START})$$

Each frequency index corresponds to a multiple of this increment. To obtain a finer frequency resolution, maximize the duration of the time window.

2. FMIN and FMAX have no effect on the *.ft0*, *.ft1*, ..., *.ftn* files.

For further information on the .FFT statement, see [Chapter 25, Performing FFT Spectrum Analysis](#).

Using Data Driven PWL Sources

Data driven PWL sources allow the results of an experiment or a previous simulation to provide one or more input sources for a transient simulation.

Syntax

```
source n+ n- PWL (param1, param2)
.DATA dataname
param1 param2
val    val
val    val
...    ...
.ENDDATA
.TRAN DATA=dataname
```

where

<i>source</i>	name of the PWL source
n+ n-	positive and negative nodes on the source
PWL	piecewise linear source keyword
<i>param1</i>	parameter name for a value provided in a .DATA statement
<i>param2</i>	parameter name for a value provided in a .DATA statement
.DATA	inline data statement
dataname	name of the inline .DATA statement
val	value for param1 or param2

The named PWL source must be used with a .DATA statement that contains *param1-param2* value pairs. This source allows the results of one simulation to be used as an input source in another simulation. The transient analysis must be data driven.

Example

```
vin in 0 PWL(timea, volta)
.DATA timvol
0      0
1      5
20     5
21     0
.enddata
```

In the example above, *vin* is a piecewise linear source with an output of 0 V from time 0 to 1 ns, 5 V from time 1 ns to time 20 ns, and 0 V from time 21 ns onward.

Always make sure the start time (time=0) is specified in the data driven analysis statement, to ensure that the stop time is calculated correctly.

